
ESTRUCTURAS SECUENCIALES 
 
 
Proceso Bienvenida 
 //Nos da la bienvenida dado tu nombre 
 Escribir "¿Como te llamas:?"; 
 Leer nombre; 
 Escribir "Bienvenido ", nombre; 
FinProceso 
 
 
 
Proceso Cilindro 
 Escribir "Radio Cilindro:"; 
 Leer R; 
 Escribir "Altura:";  
 Leer H; 
 AREA<-2*3.14*R^2+2*3.14*R*H; 
 Volumen<-3.14*R^2*H; 
 Escribir "Volumen es:",Volumen; 
 Escribir "Area es:",AREA; 
FinProceso 
 
 
 
 
 
 
 
 
 
 
 
 

ESTRUCTURAS SELECTIVAS: 
 
 
Proceso Negativo_positivo_cero 
 Escribir "Valor del número:"; 
 Leer N; 
 Si N=0 Entonces 
  Escribir "Es nulo"; 
 Sino 
  Si N>0 Entonces 
   Escribir "Es positivo"; 
  Sino 
   Escribir "Es negativo"; 
  FinSi 
 FinSi 
FinProceso 

 
 
 
 
 
 
 
 
 
 



Proceso Bisiesto 
 //Los bisiestos son divisibles por 4 salvo 
 //los fin de siglo no divisibles por 400 
 Escribir "Indica el año:"; 
 Leer N; 
 bisiesto<-Verdadero; 
 Si n mod 4 = 0 Entonces 
  Si n mod 100 = 0 Y n mod 400 <>0 Entonces 
   bisiesto<-Falso; 
  FinSi 
 Sino 
  bisiesto<-Falso; 
 FinSi 
 Si bisiesto Entonces 
  Escribir "es bisiesto"; 
 Sino 
  Escribir "No es bisiesto"; 
 FinSi 
FinProceso 
 
 

 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 



Proceso ECUACIONSEGUNDOGRADO 
  
Escribir "Coeficiente A:"; 
 Leer A;  
 Escribir "Coeficiente B:"; 
 Leer B; 
 Escribir "Coeficiente C:"; 
 Leer C; 
 //Calculamos el discriminante 
 D<-B^2-4*A*C; 
 Si D<0 Entonces 
  Escribir "No hay sol. real"; 
 Sino 
  Si D=0 Entonces 
   X1<--B/(2*A); 
   Escribir "Solucion doble:",X1; 
  Sino 
   X1<-(-B+RAIZ(D))/(2*A); 
   X2<-(-B-RAIZ(D))/(2*A); 
   Escribir "Una Solucion:",X1," Segunda Solucion", X2; 
  FinSi 
 FinSi 
FinProceso 
 

 
 
 
 
 
 
 
 
 
 
 
 
 



Proceso SelectivaMultiple 
 Escribir "1.- Opción 1"; 
 Escribir "2.- Opción 2"; 
 Escribir "3.- Opción 3"; 
 Escribir "Eliga una opcion"; 
 Leer x; 
 Segun x Hacer 
  1: 
   Escribir "Ha pulsado 1"; 
  2: 
   Escribir "Ha pulsado 2"; 
    
  3: 
   Escribir "Ha pulsado 3"; 
    
  De Otro Modo: 
   Escribir "No ha elegido opcion"; 
 FinSegun 
FinProceso 
 
 

 
 
 
 

ESTRUCTURAS REPETITIVAS 
 
 
 
Proceso BucleMientras 
 x<-1; 
 Mientras x<=100 Hacer 
  Escribir x; 
  x<-x+1; 
 FinMientras 
FinProceso FinProceso 
 
 
 



Proceso BucleRepetir  
 x<-1; 
 Repetir 
  Escribir x; 
  x<-x+1; 
 Hasta Que x>100 
  
FinProceso 
 
 
 
 
  
 
 
 
 
 
Proceso BuclePara 
 Para x<-1 Hasta 100 Con Paso 1 Hacer 
  Escribir x; 
 FinPara 
FinProceso  
 
 
 
 
 
 
 
 
 
 
Proceso Fibonacci 
 Escribir "Numero de terminos"; 
 Leer n; 
 primero<-0; 
 segundo<-1; 
 Escribir primero; 
 Para x<-2 Hasta n Con Paso 1 Hacer 
  Escribir segundo; 
  reserva<-segundo; 
  segundo<-primero+segundo; 
  primero<-reserva; 
 FinPara 
FinProceso 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



Proceso tablasmultiplicar  
 Para j<-1 Hasta 10 Con Paso 1 Hacer 
  Para i<-1 Hasta 10 Con Paso 1 Hacer 
   Escribir j,"*",i,"=",j*i; 
  FinPara 
 FinPara 
FinProceso 
 
 
 
 
 
 
 
 
 
 
 
Proceso Divisores 
 Escribir "Dame un numero entero"; 
 Leer n; 
 Para j<-1 Hasta n/2 Con Paso 1 Hacer 
  Si n mod j = 0 Entonces 
   Escribir j; 
  FinSi 
 FinPara 
FinProceso  
 
 
 
 
 
 
 
 
 
Proceso NumeroPrimo 
 Escribir "dame numero entero"; 
 Leer n;  
 primo<-verdadero; 
 Para j<-2 Hasta n/2 Con Paso 1 Hacer 
  Si n mod j = 0 Entonces 
   primo<-Falso; 
  FinSi 
 FinPara 
 Si primo Entonces 
  Escribir "Es primo"; 
 Sino 
  Escribir "No es primo"; 
 FinSi 
FinProceso 


